
* РЭС-60 — реле с учетом дребезга контактов


* ПАСПОРТ: РС4.569.435-04, РС4.569.435-09


* рабочее напряжение (3,5...4,5) В.


.subckt PC-60 coila coilb no1 nc1 com1 no2 nc2 com2


+ PARAMS:


+ T_make = 3.5mSec ; время срабатывания


+ T_break = 1.5mSec ; время отпускания


+ T_bounce =0.5mSec ; время дребезга


+ I_pull = 60ma ; ток срабатывания


+ I_drop = 13ma ; ток отпускания


+ R_coil = 35 ; сопротивление катушки постоянному току


+ L_coil = 5mH ; индуктивность катушки


+ R_open = 100MEG ; минимальное сопротивление


* разомкнутого контакта


+ R_close = .05 ; максимальное сопротивление


* замкнутого контакта


+ Contact_match = .95 ; коэффициент времени: 1 = точно





* электрическая модель катушки


v_winding coila  a1     0


r_winding a1     a2     {R_coil}


l_winding a2     coilb  {L_coil}





* преобразователь ток—напряжение


e_cc  cc 0 value = {limit(I(v_winding),-3*I_pull,3*I_pull)}


r_cc  cc 0 1k





* моделирование гистерезиса


o_mag  cc 0 relay_1 DGTLNET=d digio_1


u_dly1 buf dpwr dgnd d cn_no1 relay_2a digio_1


u_inv1 inv dpwr dgnd d cn_nc1 relay_2a digio_1


u_dly2 buf dpwr dgnd d cn_no2 relay_2b digio_1


u_inv2 inv dpwr dgnd d cn_nc2 relay_2b digio_1





* моделирование дребезга контактов


x_no1  cn_no1 com1 no1 RELAY_CONTACT_BOUNCE


+ PARAMS:


+ T_make =  {T_make}


+ T_break=  {T_break}


+ T_bounce= {T_bounce}


+ R_open =  {R_open}


+ R_close=  {R_close}


x_nc1  cn_nc1 com1 nc1 RELAY_CONTACT_BOUNCE


+ PARAMS:


+ T_make =  {T_make}


+ T_break=  {T_break}


+ T_bounce= {T_bounce}


+ R_open =  {R_open}


+ R_close=  {R_close}


x_no2  cn_no2 com2 no2 RELAY_CONTACT_BOUNCE


+ PARAMS:


+ T_make =  {T_make*Contact_match}


+ T_break=  {T_break*Contact_match}


+ T_bounce= {T_bounce*Contact_match}


+ R_open =  {R_open}


+ R_close=  {R_close}


x_nc2  cn_nc2 com2 nc2 RELAY_CONTACT_BOUNCE


+ PARAMS:


+ T_make =  {T_make*Contact_match}


+ T_break=  {T_break*Contact_match}


+ T_bounce= {T_bounce*Contact_match}


+ R_open =  {R_open}


+ R_close=  {R_close}


.model relay_1 doutput (


+ s0name="0" s0vlo={-I_pull}   s0vhi={I_pull}


+ s1name="1" s1vlo={I_drop}    s1vhi={4*I_pull}


+ s2name="1" s2vlo={-4*I_pull} s2vhi={-I_drop}


+ timestep={T_make/1000}


+ )


.model relay_2b ugate (


+ tphlmn={T_break*Contact_match}


+ tphlmx={T_break*Contact_match}


+ tplhmn={(T_make-0.126*T_bounce)*Contact_match} 


+ tplhmx={(T_make-0.126*T_bounce)*Contact_match}


+ )


.model relay_2a ugate (


+ tphlmn={T_break}   tphlmx={T_break}


+ tplhmn={(T_make-0.126*T_bounce)}


+ tplhmx={(T_make-0.126*T_bounce)}


+ )


.model digio_1 uio


.ends








